

nlsur postestimation — Postestimation tools for nlsur

[Postestimation commands](#)
 [predict](#)
 [margins](#)
 [Remarks and examples](#)
 Also see

Postestimation commands

The following postestimation commands are available after `nlsur`:

| Command | Description |
|------------------------------|---|
| <code>estat ic</code> | Akaike's, consistent Akaike's, corrected Akaike's, and Schwarz's Bayesian information criteria (AIC, CAIC, AICC, and BIC) |
| <code>estat summarize</code> | summary statistics for the estimation sample |
| <code>estat vce</code> | variance–covariance matrix of the estimators (VCE) |
| <code>estimates</code> | cataloging estimation results |
| <code>etable</code> | table of estimation results |
| <code>forecast</code> | dynamic forecasts and simulations |
| <code>hausman</code> | Hausman's specification test |
| <code>lincom</code> | point estimates, standard errors, testing, and inference for linear combinations of coefficients |
| <code>lrtest</code> | likelihood-ratio test |
| * <code>margins</code> | marginal means, predictive margins, marginal effects, and average marginal effects |
| <code>marginsplot</code> | graph the results from margins (profile plots, interaction plots, etc.) |
| <code>nlcom</code> | point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients |
| <code>predict</code> | fitted values, residuals, etc. |
| <code>predictnl</code> | point estimates, standard errors, testing, and inference for generalized predictions |
| <code>test</code> | Wald tests of simple and composite linear hypotheses |
| <code>testnl</code> | Wald tests of nonlinear hypotheses |

*You must specify the `variables()` option with `nlsur`.

predict

Description for predict

`predict` creates a new variable containing predictions such as fitted values and residuals.

Menu for predict

Statistics > Postestimation

Syntax for predict

```
predict [type] newvar [if] [in] [, equation(#eqno) yhat residuals]
```

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

Options for predict

Main

`equation(#eqno)` specifies to which equation you are referring. `equation(#1)` would mean that the calculation is to be made for the first equation, `equation(#2)` would mean the second, and so on.

If you do not specify `equation()`, results are the same as if you had specified `equation(#1)`. `yhat`, the default, calculates the fitted values for the specified equation.

`residuals` calculates the residuals for the specified equation.

margins

Description for margins

`margins` estimates margins of response for fitted values.

Menu for margins

Statistics > Postestimation

Syntax for margins

```
margins [marginlist] [, options]
margins [marginlist] , predict(statistic ...) [options]
```

| <i>statistic</i> | Description |
|------------------------|---------------------------------------|
| <code>yhat</code> | fitted values; the default |
| <code>residuals</code> | not allowed with <code>margins</code> |

Statistics not allowed with `margins` are functions of stochastic quantities other than $e(b)$.

For the full syntax, see [R] [margins](#).

Remarks and examples

[stata.com](http://www.stata.com)

▶ Example 1

In [example 2](#) of [R] [nlsur](#), we fit a four-factor translog cost function to data for the U.S. economy. The own-price elasticity for a factor measures the percentage change in its usage as a result of a 1% increase in the factor's price, assuming that output is held constant. For the translog production function, the own-price factor elasticities are

$$\eta_i = \frac{\delta_{ii} + s_i(s_i - 1)}{s_i}$$

Here we compute the elasticity for capital at the sample mean of capital's factor share. First, we use `summarize` to get the mean of `s_k` and store that value in a scalar:

```
. use https://www.stata-press.com/data/r18/mfgcost
(Manufacturing cost)
. nlsur (s_k = {bk} + {dkk}*ln(pk/pm) + {dkl}*ln(pl/pm) + {dke}*ln(pe/pm))
> (s_l = {bl} + {dkl}*ln(pk/pm) + {dll}*ln(pl/pm) + {dle}*ln(pe/pm))
> (s_e = {be} + {dke}*ln(pk/pm) + {dle}*ln(pl/pm) + {dee}*ln(pe/pm)),
> ifgnls
(output omitted)
. summarize s_k
```

| Variable | Obs | Mean | Std. dev. | Min | Max |
|----------|-----|---------|-----------|--------|--------|
| s_k | 25 | .053488 | .0044795 | .04602 | .06185 |

```
. scalar kmean = r(mean)
```

Now, we can use `nlcom` to calculate the elasticity:

```
. nlcom (([dkk]_cons + kmean*(kmean-1)) / kmean)
      _nl_1: ([dkk]_cons + kmean*(kmean-1)) / kmean
```

| | Coefficient | Std. err. | z | P> z | [95% conf. interval] | |
|-------|-------------|-----------|-------|-------|----------------------|-----------|
| _nl_1 | -.3952986 | .1083535 | -3.65 | 0.000 | -.6076676 | -.1829295 |

If the price of capital increases by 1%, its usage will decrease by about 0.4%. To maintain its current level of output, a firm would increase its usage of other inputs to compensate for the lower capital usage. The standard error reported by `nlcom` reflects the sampling variance of the estimated parameter $\widehat{\delta}_{kk}$, but `nlcom` treats the sample mean of `s_k` as a fixed parameter that does not contribute to the sampling variance of the estimated elasticity.

◀

Also see

[R] [nlsur](#) — Estimation of nonlinear systems of equations

[U] [20 Estimation and postestimation commands](#)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

